



Решавање електричног кола са операционим појачавачем




Задатак: Одредити појачање

- Вредности елемената (параметри) електричног кола са слике су познати.
- Нацртати граф кола. Одредити број главних (фундаменталних) пресека и број главних петљи (контура).
- Одредити излазни напон операционог појачавача (ОП, OpAmp).
- Одредити напонско појачање (amplification, gain) као количник излазног напона ОП и напона напонског извора (генератора).




Изрази у загради су синоними или енглески називи.


Додати четврти крај ОП




Обележити чворове




Уочити напоне (потенцијале) чворова




Усвојити смерове струја



Обележити струје отпорника




Обележити остале струје




Поступак решавања кола

- Испитати да ли је граф кола **повезан**. Ако није, онда у свакој повезаној компоненти графа усвојити по један чвор и спојити усвојене чворове.
- Нумерисати чворове узастопним целим бројевима почев од **0**. Чвор означен са 0 је упоредни чвор (нулти чвор, маса, референтни чвор, уземљење).
- Написати једначине **Кирхофовог закона за струје** (КЗС) за све чворове осим нултог (упоредног) чвора.
- Дефинисати **напоне чворова** (потенцијале чворова) као напоне између посматраног чвора и нултог чвора.
- Изразити струје приступа, ако је могуће, **преко** напона чворова.
- Ако струју приступа није могуће изразити преко напона чворова, написати **једначину елемента** (карактеристику) тог приступа.
- Добијени систем једначина су једначине **модификоване нодалне анализе** (МНА).
- **Променљиве** МНА су напони чворова и струје приступа које не могу да се израз преко напона чворова.



Modified Nodal Analysis (MNA)

Граф електричног кола



Др Дејан В. Тошић, редовни професор Теорија електричних кола, предавања

Универзитет у Београду - Електротехнички факултет, 2019.

1 КЗС

Написати једначине Кирхофовог закона за струје (КЗС) за све чворове осим нултог (упоредног) чвора.

```
(%i1) KZS: [i5 + i4 = 0, -i4 + i6 = 0,
           -i6 + i1 + i2 = 0, i7 - i2 - i3 = 0];
(KZS) [i5+i4=0, i6-i4=0, -i6+i2+i1=0, i7-i3-i2=0]
```

2 Изразити струје приступа, ако је могуће, преко напона чворова

```
(%i2) zamena: [i4 = (v4 - v1)/R4, i2 = (v2 - v3)/R2,
              i1 = (v2 - 0)/R1, i3 = (0 - v3)/R3];
(zamena) [i4 = (v4-v1)/R4, i2 = (v2-v3)/R2, i1 = v2/R1, i3 = -v3/R3]
```

```
(%i3) VKZS: ev(KZS, zamena);
(VKZS) [ (v4-v1)/R4 + i5 = 0, i6 - (v4-v1)/R4 = 0, (v2-v3)/R2 + v2/R1 - i6 = 0, v3/R3 - (v2-v3)/R2 + i7 =
0 ]
```

3 Ако струју приступа није могуће изразити преко напона чворова, написати једначину елемента тог приступа

```
(%i4) JE: [v4 = ug, i6 = 0, v1 - v2 = 0];
(JE) [v4=ug, i6=0, v1-v2=0]
```

4 МНА

Добијени систем једначина су једначине модификоване нодалне анализе (МНА), Modified Nodal Analysis (MNA).

```
(%i5) jednacineMNA: append(VKZS, JE);
(jednacineMNA) [ (v4-v1)/R4 + i5 = 0, i6 - (v4-v1)/R4 = 0, (v2-v3)/R2 + v2/R1 - i6 = 0, v3/R3 - (v2-v3)/R2 + i7 =
0, v4=ug, i6=0, v1-v2=0 ]
```

5 Решити једначине МНА

Решити електрично коло решавањем једначина МНА, одредити одзив електричног кола.

```
(%i6) odziv: linsolve(jednacinemNA, [v1,v2,v3,v4,i5,i6,i7]);
(odziv) [v1=ug, v2=ug, v3=  $\frac{R2\ ug + R1\ ug}{R1}$ , v4=ug, i5=0, i6=0, i7= -  

 $\frac{R3\ ug + R2\ ug + R1\ ug}{R1\ R3}$  ]
```

6 Одредити појачање

Одредити напонско појачање (amplification, gain) као количник излазног напона ОП и напона напонског извора (генератора).

```
(%i7) A: ev(v3/ug, odziv, ratsimp);
(A)  $\frac{R2 + R1}{R1}$ 
```

```
(%i8) expand(A);
(%o8)  $\frac{R2}{R1} + 1$ 
```

7 Одредити излазни напон ОП

```
(%i9) uIzlazno: ev(v3, odziv, factor);
(uIzlazno)  $\frac{(R2 + R1)\ ug}{R1}$ 
```

8 Одредити снагу (тренутну улазну снагу) операционог појачавача

```
(%i10) pOPizraz: (v1-v2)*i6 + (v3-0)*i7;
(pOPizraz) i7 v3 + i6 (v1 - v2)
```

```
(%i11) pOPzamena: ev(pOPizraz, zamena);
(pOPzamena) i7 v3 + i6 (v1 - v2)
```

```
(%i12) pOP: ev(pOPzamena, odziv);
(pOP) -  $\frac{(R2\ ug + R1\ ug)(R3\ ug + R2\ ug + R1\ ug)}{R1^2\ R3}$ 
```

```
(%i13) factor(pOP);
(%o13) -  $\frac{(R2+R1)(R3+R2+R1)ug^2}{R1^2 R3}$ 
```

9 Одредити снагу (тренутну улазну снагу) отпорника R3

```
(%i14) pR3izraz: (0-v3)*i3;
(%o14) -i3 v3
```

```
(%i15) pR3zamena: ev(pR3izraz, zamena);
(%o15)  $\frac{v3^2}{R3}$ 
```

```
(%i16) pR3: ev(pR3zamena, odziv, factor);
(%o16)  $\frac{(R2+R1)^2 ug^2}{R1^2 R3}$ 
```

10 Одредити струју извора

Одредити струју напонског извора (идеалног напонског генератора).

```
(%i17) iug: ev(i5, odziv);
(%o17) 0
```

11 Одредити појачање

Одредити појачање за $R1 = 10 \text{ kOhm}$, $R2 = 20 \text{ kOhm}$.

```
(%i18) kOhm: 1000 $
```

```
(%i19) Abrojno = ev(A, [R1=10*kOhm, R2=20*kOhm]);
(%o19) Abrojno = 3
```

Одредити појачање ако је однос (количник) отпорника $R2/R1$ општи број (симболичка променљива) x .

```
(%i20) Ax: ev(A, R2=x*R1, ratsimp);
(%o20) x+1
```